Applied Complex Analysis (2021)

1 Lecture 14: Inverting the Hilbert transform and ideal fluid flow

In this lecture we

1. Discuss how to invert a Hilbert transform
2. Use it to solve ideal fluid flow around a plate

1.1 Inverting the Hilbert transform

We now consider the problem of finding u such that

$$H_{[a,b]}u(x) = f(x)$$

for $a < x < b$. Note that this is an additive jump problem: if we write $\phi(z) = \mathcal{C}u(z)$ then we want to solve

$$-\mathrm{i}(\phi_+(x) + \phi_-(x)) = f(x)$$

where ϕ satisfies the conditions of Plemelj (Analyticity off $[a, b]$, weaker than pole singularities, decay at ∞). If we can find such a ϕ, Plemelj guarantees that u is recovered via

$$\phi_+(x) - \phi_-(x) = (\mathcal{C}^+ - \mathcal{C}^-)u(x) = u(x)$$

To tackle this we are going to reduce it to a subtractive problem. Writing

$$\kappa(z) = \sqrt{z-a}\sqrt{z-b}$$

which satisfies

$$\kappa_+(x) = \mathrm{i}\sqrt{b-x}\sqrt{x-a} = -\kappa_-(x)$$

Thus if we write

$$\phi(z) = \frac{\psi(z)}{\kappa(z)}$$

for some new unknown ψ we have that

$$f(x) = -\mathrm{i}(\phi_+(x) + \phi_-(x)) = -\mathrm{i}\left(\frac{\psi_+(x)}{\kappa_+(x)} + \frac{\psi_-(x)}{\kappa_-(x)}\right) = -\mathrm{i}\frac{\psi_+(x) - \psi_-(x)}{\kappa_+(x)}$$

In other words, we want

$$\psi_+(x) - \psi_-(x) = \mathrm{i}f(x)\kappa_+(x) = -f(x)\sqrt{b-x}\sqrt{x-a}$$

where we need ψ to be bounded (the decay of $1/\kappa$ then ensures the decay of ϕ). Therefore by Plemelj we have for $g(x) = f(x)\sqrt{b-x}\sqrt{x-a}$

$$\psi(z) = -\mathcal{C}g(z) - D$$

and hence for

$$\begin{aligned}
u(x) = \phi_+(x) - \phi_-(x) &= \frac{\psi_+(x)}{\kappa_+(x)} - \frac{\psi_-(x)}{\kappa_-(x)} \\
&= -\frac{\mathcal{C}^+g(x) + \mathcal{C}^-g(x) + 2D}{\kappa_+(x)} = -\frac{\mathrm{i}\mathcal{H}g(x) + 2D}{\kappa_+(x)} \\
&= -\frac{\mathcal{H}_{[a,b]}[f\sqrt{b-\diamond}\sqrt{\diamond - a}](x) + 2D}{\sqrt{b-x}\sqrt{x-a}}
\end{aligned}$$

where $\diamond$ denotes a dummy variable and D is arbitrary.

In practice we determine $\mathcal{H}g(x)$ by first computing its Cauchy transform.

Example Let's try with $f(x) = x$ on $[-1, 1]$. The first step is to compute the Cauchy transform of $g(x) = x\sqrt{1-x^2}$. Using the usual technique this gives

$$\mathcal{C}g(z) = \frac{z\sqrt{z-1}\sqrt{z+1} - z^2 + 1/2}{2\mathrm{i}}$$

where we use the Laurent series

$$\begin{aligned}
\sqrt{z-1}\sqrt{z+1} &= z\sqrt{1-1/z}\sqrt{1+1/z} \\
&= z(1 + 1/(2z) - 1/(8z^2) + O(z^{-3}))(1 - 1/(2z) - 1/(8z^2) + O(z^{-3})) \\
&= z - \frac{1}{2z} + O(z^{-2})
\end{aligned}$$

Therefore

$$\begin{aligned}
\mathcal{H}g(x) &= -\mathrm{i}(\mathcal{C}^+ + \mathcal{C}^-)g(x) = -\frac{x\sqrt{x-1}_+\sqrt{x+1} - x\sqrt{x-1}_-\sqrt{x+1}}{2} + x^2 - 1/2 \\
&= x^2 - 1/2
\end{aligned}$$

Giving us

$$u(x) = -\frac{x^2 - 1/2 + 2D}{\sqrt{1-x^2}}$$

Choosing $D = -1/4$ recovers the known solution $\sqrt{1-x^2}$.

1.2 Ideal fluid flow

Understanding branch cuts and Cauchy transforms allows us to systematically solve equations involving the Laplace equation. A classic example is ideal fluid flow. Consider the case of uniform flow with angle θ around an infinitesimally small plate on $[-1, 1]$. We can model this as

$$\begin{aligned} v(x,y) &\sim y\cos\theta - x\sin\theta \qquad \text{as} \qquad x^2+y^2 \to \infty \\ v_{xx} + v_{yy} &= 0 \\ v(x,0) &= 0 \qquad \text{for} \qquad -1 < x < 1 \end{aligned}$$

Using the techniques we developed in the last few lectures, we obtain a nice, simple, closed form expression for the solution as the imaginary part of an analytic function:

```
using Plots, LinearAlgebra, ApproxFun, SingularIntegralEquations
Cw = (θ,z) -> -im*sin(θ)*(sqrt(z-1)sqrt(z+1) - z)
φ = (θ,z) -> exp(-im*θ)*z + Cw(θ,z)
u = (θ,x,y) -> imag(φ(θ, x + im*y))

xx = yy = range(-3; stop=3 , length=500)
contour(xx, yy, u.(1.3,xx',yy); nlevels = 100)
plot!(Segment(-1.,1.); color=:black, label="obstacle")
```



We divide this task into stages:

1. Rephrasing as a complex-analytical problem: $v(x, y)$ to $\phi(z)$
2. Reduction to inverting a Hilbert transform: $\phi(z)$ to $w(x)$
3. Calculating the inverse Hilbert transform: Finding $w(x)$
4. Calculating its Cauchy transform: $w(x)$ to $\phi(z)$

1.2.1 Real and imaginary parts of analytic functions satisfy Laplace's equation

The real and imaginary parts of an analytic function satisfy Laplace's equation: that is if $\phi(z) = \phi(x + \mathrm{i}y) = u(x, y) + \mathrm{i}v(x, y)$ where u and v are the real/imaginary parts, then

$$
\begin{aligned}
u_{xx} + u_{yy} &= 0 \\
v_{xx} + v_{yy} &= 0
\end{aligned}
$$

To see this, note that the complex-derivative of $\phi(z)$ can be written in terms of two different partial derivatives:

$$
\phi'(z) = \lim_{h\to 0} \frac{\phi(z+h) - \phi(z)}{h} = \lim_{h\to 0} \frac{u(x+h,y) - u(x,y) + \mathrm{i}(v(x+h,y) - v(x,y))}{h} = u_x + \mathrm{i}v_x
$$
$$
\phi'(z) = \lim_{h\to 0} \frac{\phi(z+\mathrm{i}h) - \phi(z)}{\mathrm{i}h} = \lim_{h\to 0} \frac{u(x,y+h) - u(x,y) + \mathrm{i}(v(x,y+h) - v(x,y))}{\mathrm{i}h} = -\mathrm{i}u_y + v_y
$$

Taking a second derivative we get two equations:

$$
\phi''(z) = u_{xx} + \mathrm{i}v_{xx} = -u_{yy} - \mathrm{i}v_{yy}
$$

which implies $u_{xx} + u_{yy} = 0$ and $v_{xx} + v_{yy} = 0$.

1.2.2 2. Reduce PDE to the Hilbert transform of an unknown function

Therefore we can rewrite the ideal fluid flow equation as a problem of calculating $\phi(z) = u(x, y) + \mathrm{i}v(x, y)$ whose imaginary part is the solution to the ideal fluid flow PDE (we don't use the real part u). That is, we want to find analytic $\phi(z)$ that satisfies

$$
\begin{aligned}
\phi(z) &\sim \mathrm{e}^{-\mathrm{i}\theta} z \qquad \text{as} \qquad z \to \infty \\
\Im\phi(x) &= 0 \qquad \text{for} \qquad -1 < x < 1
\end{aligned}
$$

Write

$$
\phi(z) = \mathrm{e}^{-\mathrm{i}\theta} z + c + \mathcal{C}_{[-1,1]} w(z)
$$

for an as-of-yet unknown function w and c an unknown constant, we have (assuming w is real) that

$$
0 = \Im\phi(x) = -x \sin\theta + \Im c + \Im \mathcal{C}^{+}_{[-1,1]} w(x) = -x\sin\theta + \Im c + \frac{1}{2}\mathcal{H}w(x)
$$

In this example, we can take $c = 0$. Therefore, we want to solve

$$
\mathcal{H}w(x) = 2x \sin\theta
$$

for w.

1.2.3 3. Calculating the inverse Hilbert transform

We now plug the problem into the inverse Hilbert transform formula:

$$w(x) = \frac{-1}{\sqrt{1-x^2}} \mathcal{H}[f\sqrt{1-\diamond^2}](x) + \frac{D}{\sqrt{1-x^2}}$$

where $f(x) = 2x\sin\theta$. As we found before,

$$\mathcal{C}[\diamond\sqrt{1-\diamond^2}](z) = \frac{z\sqrt{z-1}\sqrt{z+1} - z^2 + 1/2}{2\mathrm{i}}$$

and therefore

$$\mathcal{H}[\diamond\sqrt{1-\diamond^2}](x) = -\mathrm{i}(\mathcal{C}^+ + \mathcal{C}^-)[\diamond\sqrt{1-\diamond^2}](x) = x^2 - \frac{1}{2}$$

Thus (relabeling D) we have

$$w(x) = 2\sin\theta \frac{D - x^2}{\sqrt{1-x^2}}$$

Demonstration Here we see that this gives us the right Hilbert transform:

```
D = randn()
θ = 0.1
x = Fun()
w = 2sin(θ) * (D-x^2)/sqrt(1-x^2)
-hilbert(w,0.2) , 2sin(θ)*0.2 # Minus in front of w to fix normalisation
```

```
(0.03993336665873126, 0.03993336665873126)
```

$$D$$

is arbitrary, but from physical principles we know that we don't want the solution to blow up. If w blows up then so does its Cauchy transform. Therefore, we choose $D = 1$ so that

$$w(x) = 2\sin\theta\sqrt{1-x^2}$$

1.2.4 4. Calculating its Cauchy transform

Now recall

$$\mathcal{C}\left[\sqrt{1-\diamond^2}\right](z) = \frac{\sqrt{z-1}\sqrt{z+1} - z}{2\mathrm{i}}$$

Therefore, $w(x) = 2\sin\theta\sqrt{1-x^2}$, implies

$$\mathcal{C}w(z) = -\mathrm{i}\sin\theta(\sqrt{z-1}\sqrt{z+1} - z)$$

which means

$$\phi(z) = \mathrm{e}^{-\mathrm{i}\theta}z - \mathrm{i}\sin\theta(\sqrt{z-1}\sqrt{z+1} - z).$$

1.3 Numerical example of two intervals

We note that for obstacles on the real line, represented by a contour Γ, the problem of ideal fluid flow around Γ is still reducible to solving the singular integral equation

$$\mathcal{H}_\Gamma f(x) = 2x \sin\theta$$

Even when not solvable exactly, one can solve it numerically:

```
a = 0.3
θ = 1.3
Γ = Segment(-1,-a) ∪ Segment(a, 1)

x = Fun(Γ)
sp = PiecewiseSpace(JacobiWeight.(0.5,0.5,components(Γ))...)
H = Hilbert(sp)

o_1 = Fun(x -> -1 ≤ x ≤ -a ? 1 : 0, Γ )
o_2 = Fun(x -> a ≤ x ≤ 1 ? 1 : 0, Γ )

a, b, f = [o_1 o_2 H] \ [-2x*sin(θ)]

φ = (θ,z) -> exp(-im*θ)*z + cauchy(f, z)
u = (θ,x,y) -> imag(φ(θ, x + im*y))

xx = yy = range(-3.; stop=3., length=500)
contour(xx, yy, u.(θ, xx',yy); nlevels = 100)
plot!(Γ; color=:black, label="obstacle")
```