1 Lecture 15: Electrostatic charges in a potential well

1. One charge in a well
2. Two charges in a well
3. N charges in a well
4. Limiting distribution

We study the dynamics of many electric charges in a potential well. We restrict our attention to 1D: picture an infinitely long wire with charges on it. We will see that as the number of charges becomes large, we can determine the limiting distribution by inverting the Hilbert transform, but one for which the *interval* it is posed on is solved for.

1.1 One charge in a potential well

Consider a point charge in a potential well $V(x) = x^2/2$, initially located at λ^0. The dynamics of the point charge are governed by Newton's law

$$m\frac{\mathrm{d}^2\lambda}{\mathrm{d}t^2} + \gamma\frac{\mathrm{d}\lambda}{\mathrm{d}t} = -V'(\lambda) = -\lambda$$

Here m is the mass (we will always take it to be 1) and γ is a damping constant (think friction) which ensures that the charge eventually comes to rest.

This is a simple damped harmonic oscillator: if we are positive we move left and if we are negative we move right. Here we show the evolution as a function of time, where we plot this on top of the potential to see its effect.

```
using DifferentialEquations, Plots, StaticArrays
V = x -> x^2/2  # Potential
Vp = x -> x     # Force

λ_0 = 2.3  # initial location
v_0 = 1.2  # initial velocity
γ = 1.0    # damping

λv = solve(ODEProblem(((λ,v),_,t) -> SVector(v,-Vp(λ) - γ*v), SVector(λ_0,v_0), (0.0,
20.0)); reltol=1E-6);

tt = range(0.,20; length=1000)
xx = range(-sqrt(2*last(tt)),sqrt(2*last(tt)); length=1000)
plot(xx, V.(xx); label="potential", xlabel="x", ylabel="time and potential")
plot!(first.(λv.(tt)), tt; label="charge")
```



In the limit the charge reaches an equilibrium: it no longer varies in time. I.e., it reaches a point where $\frac{\mathrm{d}\lambda^2}{\mathrm{d}t^2} = \frac{\mathrm{d}\lambda}{\mathrm{d}t} = 0$, which is equivalent to solving

$$0 = -V'(\lambda) = -\lambda$$

in other words, the minimum of the well, in this case $\lambda = 0$.

1.2 Two charges in a potential well

A point charge at x_0 in 2D creates a (Newtonian) potential field of $V_{x_0}(x) = -\log|x - x_0|$ with derivative

$$V'_{x_0}(x) = \begin{cases} -\frac{1}{x-x_0} & x > x_0 \\ \frac{1}{x_0-x} & x < x_0 \end{cases} = \frac{1}{x_0 - x}$$

Here we are thinking of the charges as 2D but restricted to the real line: think of a 1D wire embedded in 2D space.

Suppose there are now two charges, λ_1 and λ_2 but no potential well. The effect on the first charge λ_1 is to repulse away from λ_2 via the potential field $V_{\lambda_2}(x) = -\log|x - \lambda_2|$. Since we have

$$m\frac{\mathrm{d}^2\lambda_1}{\mathrm{d}t^2} = -V'_{\lambda_2}(\lambda_1) = \frac{1}{\lambda_1 - \lambda_2}$$

Similarly, the effect on λ_2 is

$$m\frac{\mathrm{d}^2\lambda_2}{\mathrm{d}t^2} = \frac{1}{\lambda_2 - \lambda_1}$$

Unrestricted, the two potentials will repulse off to infinity:

```
N = 2
λ_0 = randn(2)  # random initial location
v_0 = randn(2)  # random initial velocity
prob = ODEProblem(function(λv,_,t)
        λ = λv[1:N]
        v = λv[N+1:end]
        [v; 1/(λ[1] - λ[2]); 1/(λ[2] - λ[1])]
        end, [λ_0; v_0], (0.0, 10.0))
λv = solve(prob; reltol=1E-6)
tt = range(0.,20; length=1000)
λv(0.1)

p = plot(;xlabel="x",ylabel="t")
for j = 1:N
    plot!(getindex.(λv.(tt),j), tt; label="charge $j")
end
p
```



Adding in a potential well and damping and we get an equilibrium again:

$$\frac{\mathrm{d}^2\lambda_1}{\mathrm{d}t^2} + \gamma\frac{\mathrm{d}\lambda_1}{\mathrm{d}t} = \frac{1}{\lambda_1 - \lambda_2} - V'(\lambda_1)$$
$$\frac{\mathrm{d}^2\lambda_2}{\mathrm{d}t^2} + \gamma\frac{\mathrm{d}\lambda_2}{\mathrm{d}t} = \frac{1}{\lambda_2 - \lambda_1} - V'(\lambda_2)$$

which we see here

```
V = x -> x^2/2  # Potential
Vp = x -> x     # Force
γ = 1.0     # damping

prob = ODEProblem(function(λv,_,t)
        λ = λv[1:N]
        v = λv[N+1:end]
        [v; 1/(λ[1] - λ[2]) - Vp(λ[1]) - γ*v[1]; 1/(λ[2] - λ[1])  - Vp(λ[2]) - γ*v[2]]
        end, [λ_0; v_0], (0.0, 20.0))
λv = solve(prob; reltol=1E-6);

xx = range(-sqrt(2*last(tt)),sqrt(2*last(tt)); length=1000)
p = plot(xx, V.(xx); label="potential", xlabel="x", ylabel="time and potential")
for j = 1:N
    plot!(getindex.(λv.(tt),j), tt; label="charge $j")
end
p
```



The fixed point is when the change with time vanishes, given by

$$
\begin{aligned}
0 &= \frac{1}{\lambda_1 - \lambda_2} - V'(\lambda_1) \\
0 &= \frac{1}{\lambda_2 - \lambda_1} - V'(\lambda_2)
\end{aligned}
$$

For this potential, we can solve it exactly: we need to solve

$$\lambda_1 = \frac{1}{\lambda_1 - \lambda_2}$$
$$\lambda_2 = \frac{1}{\lambda_2 - \lambda_1}$$

Using $\lambda_1 = -\lambda_2$, we find that $\lambda_1 = \pm\frac{1}{\sqrt{2}}$.

1.2.1 N charges in a potential well

We now consider N charges, where each charge repulses every other charge by a logarithmic potential, so we end up needing to sum over all the other charges:

$$m\frac{\mathrm{d}^2\lambda_k}{\mathrm{d}t^2} + \gamma\frac{\mathrm{d}\lambda_k}{\mathrm{d}t} = \sum_{\substack{j=1 \\ j\neq k}}^{N} \frac{1}{\lambda_k - \lambda_j} - V'(\lambda_k)$$

```
N = 100

V = x -> x^2/2   # Potential
Vp = x -> x      # Force

λ_0 = randn(N)  # initial location
v_0 = randn(N)  # initial velocity
γ = 1.0     # damping

prob = ODEProblem(function(λv,_,t)
        λ = λv[1:N]
        v = λv[N+1:end]
        [v; [sum(1 ./ (λ[k] .- λ[[1:k-1;k+1:end]])) - Vp(λ[k]) -  γ*v[k] for k=1:N]]
        end, [λ_0; v_0], (0.0, 20.0))
λv = solve(prob; reltol=1E-6)

xx = range(-sqrt(2*last(tt)),sqrt(2*last(tt)); length=1000)
p = plot(xx, V.(xx); legend=false, xlabel="x", ylabel="time and potential")
for j = 1:N
    plot!(getindex.(λv.(tt),j), tt; color=:red)
end
p
```



As the number of charges becomes large, they spread off to infinity. In the case of $V(x) = x^2$, we can renormalize by dividing by N so they stay bounded: $\mu_k = \frac{\lambda_k}{\sqrt{N}}$.

```
p = plot(xx, V.(xx); legend=false, xlabel="x", ylabel="time and potential")
for j = 1:N
    plot!(getindex.(λv.(tt)./sqrt(N),j), tt; color=:red)
end
p
```



This begs questions: why does it balance out at $\pm\sqrt{2}$? Why does it have a nice histogram precisely like $\frac{\sqrt{2-x^2}}{\pi}$:

```
histogram(λv.(20.0)[1:N]./sqrt(N); nbins=20, normalize=true, label="histogram of
charges")
plot!(x -> sqrt(2-x^2)/(π), range(eps()-sqrt(2.0); stop=sqrt(2)-eps(), length=100),
label="semicircle")
```



1.2.2 Equilibrium distribution

Plugging in $\lambda_k = \sqrt{N}\mu_k$, we get a dynamical system for μ_k:

$$m\frac{\mathrm{d}^2\mu_k}{\mathrm{d}t} + \gamma\frac{\mathrm{d}\mu_k}{\mathrm{d}t} = \frac{1}{N}\sum_{\substack{j=1\\ j\neq k}}^{N}\frac{1}{\mu_k - \mu_j} - \mu_k$$

(The choice of scaling like $\sqrt{N}$ was dictated by $V(x)$, if $V(x) = x^4$ it would be $N^{1/4}$.) Thus the limit of the charges is given when the change with t vanishes, that is

$$0 = \frac{1}{N}\sum_{\substack{j=1\\ j\neq k}}^{N}\frac{1}{\mu_k - \mu_j} - \mu_k, \qquad k = 1, \ldots, N$$

It is convenient to represent the point charges by Dirac delta functions:

$$w_N(x) = \frac{1}{N}\sum_{k=1}^{N}\delta_{\mu_k}(x)$$

normalized so that $\int w_N(x)\mathrm{d}x = 1$, so that

$$\frac{1}{N}\sum_{j=1}^{N}\frac{1}{x - \mu_j} = \int_{-\infty}^{\infty}\frac{w_N(t)\mathrm{d}t}{x - t}$$

or in other words, we have

$$\mathcal{H}_{(-\infty,\infty)} w_N(\mu_k) = \frac{V'(\mu_k)}{\pi}, \qquad k = 1, \ldots, N$$

since

$$\mathcal{H} w_N(\mu_k) = \frac{1}{\pi} \lim_{\epsilon \to 0} \left(\int_{-\infty}^{\mu_k - \epsilon} + \int_{\mu_k + \epsilon}^{\infty} \right) \frac{w_N(t)}{\mu_k - t} \mathrm{d}t = \frac{1}{N\pi} \sum_{j \neq k} \frac{1}{\mu_k - \mu_j}$$

Formally (see a more detailed explanation below), $w_N(x)$ tends to a continuous limit as $N \to \infty$, which we have guessed from the histogram to be $w(x) = \frac{\sqrt{2-x^2}}{\pi}$ for $-\sqrt{2} < x < \sqrt{2}$. We expect this limit to satisfy the same equation as w_N, that is

$$\mathcal{H} w(x) = \frac{x}{\pi}$$

for x in the support of $w(x)$.

Why is it $[-\sqrt{2}, \sqrt{2}]$? Consider the problem posed on a general interval $[a, b]$ where a and b are unknowns. We want to choose the interval $[a, b]$ so that there exists a $w(x)$ satisfying

1. $$w$$
   is bounded (Based on observation)
2. $$w(x) \geq 0$$
   for $a \leq x \leq b$ (Since it is a probability distribution)
3. $$\int_a^b w(x) \mathrm{d}x = 1$$
   (Since it is a probability distribution)
4. $$\mathcal{H}_{[a,b]} w(x) = x/\pi$$

As we saw last lecture, there exists a bounded solution to $\mathcal{H}_{[-b,b]} u = x/\pi$, namely $u(x) = \frac{\sqrt{b^2-x^2}}{\pi}$. The choice $b = \sqrt{2}$ ensures that $\int_{-b}^{b} u(x) \mathrm{d}x = 1$, hence $u(x) = w(x)$.

1.2.3 Aside: Explanation of limit of $w_N(x)$

This is beyond the scope of the course, but the convergence of $w_N(x)$ to $w(x)$ is known as weak-* convergence. A simple version of this is that

$$\int_c^d w_N(x) \mathrm{d}x \to \int_c^d w(x) \mathrm{d}x$$

for every choice of interval (c, d). $\int_c^d w_N(x) \mathrm{d}x$ is precisely the number of charges in (c, d) scaled by $1/N$, which is exactly what a histogram plots.

```julia
using ApproxFun, SingularIntegralEquations
a = -0.1; b= 0.3;
w = Fun(x -> sqrt(2-x^2)/π, a .. b)
sum(w),  # integral of w(x) between a and b
length(filter(λ -> a ≤ λ ≤ b, λv(20.0)[1:N]/sqrt(N)))/N # integral of w_N(x) between a
and b
```

```
(0.1790059168895313, 0.18)
```

Another variant of describing weak-* convergence is that the Cauchy transforms converge, that is, for z on any contour γ surrounding a and b (now the support of w) we have

$$\int_a^b \frac{w_N(x)}{x-z}\mathrm{d}x \to \int_a^b \frac{w(x)}{x-z}\mathrm{d}x$$

converges uniformly with respect to $N \to \infty$. here we demonstrate it converges at a point:

```julia
x = Fun(-sqrt(2) .. sqrt(2))
w = sqrt(2-x^2)/π
z = 0.1+0.2im
cauchy(w, z), (sum(1 ./((λv(20.0)[1:N]/sqrt(N)) .- z))/N)/(2π*im)
```

```
(0.1949409042727309 + 0.013681505291622225im, 0.19542062466509094 + 0.01372
4291088994856im)
```