Applied Complex Analysis (2021)

1 Lecture 16: Logarithmic singular integrals

The motivation behind this lecture is to calculate electrostatic potentials. An example is the Faraday cage: imagine a series of metal plates connected together so that they have the same charge. If configured to surround a region, this configuration will shield the interior from an external charge, see figure. Here, the coloured lines are equipotential lines, and there is a point source at $x = 2$, which corresponds to a forcing of $\log \|(x, y) - (2, 0)\| = \log |z - 2|$ where $z = x + \mathrm{i}y$.



1.1 Logarithmic singular integrals

From the Green's function of the Laplacian, it is natural to consider logarithmic singular integrals over a contour

$$L_\gamma f(z) = \frac{1}{\pi} \int_\gamma f(\zeta) \log|\zeta - z| \mathrm{d}s$$

where $\mathrm{d}s$ is the arclength differential (as in line integrals, not complex analysis) and f is real-valued. We focus on contours where the arclength differential is just the standard one:

$$L_{[a,b]} f(z) = \frac{1}{\pi} \int_a^b f(t) \log|z - t| \mathrm{d}t$$

Physically, $L_{[a,b]} f(z)$ can be thought of as (the negative of) the potential at the point (x, y), with $z = x + \mathrm{i}y$, that arises from a charge density of $f(x)$ on a wire on $[a, b]$. For example, for a point charge at $x_0 \in [a, b]$, we have $f(x) = \delta(x - x_0)$ and $L_{[a,b]} f(z) = \log|z - x_0|$. Logarithmic singular integrals not only have applications in electrostatics, but also in approximation theory (see B. Fornberg, *A Practical Guide to Pseudospectral Methods* and L.N. Trefethen, *Approximation Theory and Approximation Practice*).

Note that off $[a, b]$, $u(x, y) \equiv u(x + \mathrm{i}y) = Lf(x + \mathrm{i}y)$ solves Laplace's equation, and as the integrand has an integrable singularity it is continuous on $[a, b]$:

```
using ApproxFun, SingularIntegralEquations, Plots
t = Fun()
f = sqrt(1-t^2)*exp(t)
```

```
u = z -> logkernel(f, z)  # logkernel(f,z) calculates 1/π * \int f(t)*log|t-z| dt
xx = yy = -2:0.01:2
U = u.(xx' .+ im*yy)
contour(xx, yy, U)
plot!(domain(t); color=:black, label="contour")
```



Or this can also be seen from a surface plot:

```
surface(xx, yy, U)
```



For $z \notin (-\infty, b]$ the fact that u is harmonic (solves Laplace's equation) can also be seen since u is the real part of an analytic function:

$$u(z) = \Re M f(z) \qquad \text{for} \qquad M f(z) := \frac{1}{\pi} \int_a^b f(t) \log(z - t) \mathrm{d}t$$

Note that the integrand avoids the branch cut of $\log z$. To extend this to $z \in (-\infty, a]$ (or more generally, $z \notin [a, \infty)$), we can use the alternative expression

$$u(z) = \Re \tilde{M} f(z) \qquad \text{for} \qquad \tilde{M} f(z) := \frac{1}{\pi} \int_a^b f(t) \log(t - z) \mathrm{d}t$$

which follows from $\log |z - t| = \log |t - z|$.

1.2 Evaluating logarithmic singular integrals

To evaluate $Lf(z)$ we evaluate $Mf(z)$. Following our claim that the right way to represent analytic functions is by their behaviour at singularitites/branch cuts, we will accomplish this this by investigating the jumps of $Mf(z)$. This allows us to reduce it to computing a Cauchy transform.

Lemma (Jumps of M) Suppose f satisfies the conditions of Plemelj. Then $Mf(z)$ satisfies the following

1. Analyticity: M is analytic off $(-\infty, b]$
2. Regularity: M has weaker than pole singularities
3. Asymptotics:

$$Mf(z) = \frac{1}{\pi} \int_a^b f(t)\mathrm{d}t \log z + o(1) \qquad \text{as} \qquad z \to \infty$$

4. Jump: for $x < a$ we have

$$M^+ f(x) - M^- f(x) = 2\mathrm{i} \int_a^b f(t)\mathrm{d}t$$

and for $a < x < b$ we have

$$M^+ f(x) - M^- f(x) = 2\mathrm{i} \int_x^b f(t)\mathrm{d}t$$

Proof

Analyticity property (1) follows immediately from definition: we have for $z \notin (-\infty, b]$

$$\frac{\mathrm{d}}{\mathrm{d}z} M f(z) = \frac{1}{\pi} \int_a^b f(t) \frac{1}{z-t} \mathrm{d}t = -2\mathrm{i}\mathcal{C}_{[a,b]} f(z)$$

The regularity property (2) follows from the expression as an antiderivative: we know $\mathcal{C} f(z)$ has weaker than pole singularities and integrating only makes them better. To derive the asymptotics (3) we use

$$\log(z-t) = \log z + \log(1 - t/z)$$

Finally we get to the jumps. For any point $c > b$ we write

$$M f(z) = -2\mathrm{i} \int_c^z \mathcal{C} f(\zeta) \mathrm{d}\zeta + M f(c)$$

For $x < a$ we can deform the contour just above/below the interval giving us

$$M^{\pm} f(x) = M f(c) - 2\mathrm{i} \int_c^b \mathcal{C} f(t) \mathrm{d}t - 2\mathrm{i} \int_b^a \mathcal{C}^{\pm} f(t) \mathrm{d}t - 2\mathrm{i} \int_a^x \mathcal{C} f(t) \mathrm{d}t$$

so that

$$(M^+ - M^-) f(x) = -2\mathrm{i} \int_b^a (\mathcal{C}^+ - \mathcal{C}^-) f(t) \mathrm{d}t = 2\mathrm{i} \int_a^b f(t) \mathrm{d}t$$

Similarly for $a < x < b$ we have

$$M^{\pm} f(x) = M f(c) - 2\mathrm{i} \int_c^b \mathcal{C} f(t) \mathrm{d}t - 2\mathrm{i} \int_b^x \mathcal{C}^{\pm} f(t) \mathrm{d}t$$

which gives

$$(M^+ - M^-) f(x) = -2\mathrm{i} \int_b^x (\mathcal{C}^+ - \mathcal{C}^-) f(t) \mathrm{d}t = 2\mathrm{i} \int_x^b f(t) \mathrm{d}t.$$

■

We can construct a function that satisfies the same 4 properties using the Cauchy transform. By uniqueness they must be the same.

Theorem (Log kernel as Cauchy) Suppose f satisfies the conditions of Plemelj and define the (negative) indefinite integral of f via

$$F(x) := \int_x^b f(t)\mathrm{d}t.$$

Then

$$Mf(z) = \frac{\log(z-a)}{\pi}\int_a^b f(t)\mathrm{d}t + 2\mathrm{i}\mathcal{C}_{[a,b]}F(z)$$

Proof

Define

$$\phi(z) := \frac{\log(z-a)}{\pi}\int_a^b f(t)\mathrm{d}t + 2\mathrm{i}\mathcal{C}_{[a,b]}F(z)$$

This satisfies conditions (1–4) above. Therefore $\phi(z) - Mf(z)$ is continuous hence analytic on $(-\infty, a)$ and (a, b), has weaker than pole singularities at a and b so analytic there too: it is entire. The logarithmic growth at infinity cancels therefore it is zero by Liouville.

■

1.3 Examples of logarithmic singular integrals

We look at 3 examples of computing logarithmic singular integrals: $x/\sqrt{1-x^2}$, 1, and $1/\sqrt{1-x^2}$

1.3.1 Example 1

Consider $f(x) = x/\sqrt{1-x^2}$ and we want to compute

$$Lf(z) = \frac{1}{\pi}\int_{-1}^{1} \frac{t}{\sqrt{1-t^2}} \log|t-z| \mathrm{d}t = \Re \underbrace{\frac{1}{\pi}\int_{-1}^{1} \frac{t}{\sqrt{1-t^2}} \log(z-t) \mathrm{d}t}_{Mf(z)}$$

We have

$$\frac{\mathrm{d}}{\mathrm{d}x}\sqrt{1-x^2} = -f(x)$$

hence

$$F(x) = \int_x^1 f(t)\mathrm{d}t = \sqrt{1-x^2}.$$

and in particular $F(-1) = 0$. Recall that

$$\mathcal{C}F(z) = \frac{\sqrt{z-1}\sqrt{z+1} - z}{2\mathrm{i}}$$

Thus the result from last lecture gives

$$Mf(z) = \frac{1}{\pi}\int_{-1}^{1} f(t)\mathrm{d}t \log(z+1) + 2\mathrm{i}\mathcal{C}F(z) = \sqrt{z-1}\sqrt{z+1} - z$$

Let's double check the formula:

```julia
using ApproxFun, SingularIntegralEquations, Plots
x = Fun()
f = x/sqrt(1-x^2)
Mf = z -> sqrt(z-1)sqrt(z+1) - z
Lf = z -> real(Mf(z))
z = 0.1+0.1im
logkernel(f,z),Lf(z)
```

```
(-0.09000049991252067, -0.09000049991252063)
```

Here is a plot of the solution:

```julia
xx = range(-3,3;length=300); yy = range(-2,2;length=200)
Z = Lf.(xx' .+ im*yy)
contour(xx,yy,Z;ratio=1.0)
plot!(-1..1; color=:black, label="contour")
```


contour
2
1
0
−1
−2
−3
−2
−1
0
1
2
3
0.75
0.50
0.25
0
−0.25
−0.50
−0.75

1.3.2 Example 2

Consider $f(x) = 1$ and we want to compute

$$Lf(z) = \frac{1}{\pi}\int_{-1}^{1} \log|t - z|\mathrm{d}t = \Re \underbrace{\frac{1}{\pi}\int_{-1}^{1} \log(z - t)\mathrm{d}t}_{Mf(z)}$$

We have

$$F(x) = \int_x^1 f(t)\mathrm{d}t = 1 - x.$$

We can determine the Cauchy transform by considering $1 - z$ times the known Cauchy transform

$$\mathcal{C}1(z) = \frac{\log(z-1) - \log(z+1)}{2\pi\mathrm{i}} = \frac{\mathrm{i}}{\pi z} + O(z^{-3})$$

so that

$$\mathcal{C}F(z) = (1-z)\frac{\log(z-1) - \log(z+1)}{2\pi\mathrm{i}} + \frac{\mathrm{i}}{\pi}$$

Thus the result from last lecture gives

$$M1(z) = \frac{1}{\pi}\int_{-1}^{1} \mathrm{d}t \log(z+1) + 2\mathrm{i}\mathcal{C}F(z) = \frac{2}{\pi}\log(z+1) + (1-z)\frac{\log(z-1)-\log(z+1)}{\pi} - \frac{2}{\pi}$$
$$= \frac{(1-z)\log(z-1) + (1+z)\log(z+1) - 2}{\pi}$$

For $-1 < x < 1$ real this gives a particularly simple formula:

$$L1(x) = \Re M^{+}1(x) = \frac{(1-x)\log(1-x) + (1+x)\log(x+1) - 2}{\pi}$$

We can double check this:

```
f = Fun(1,-1..1)
x = 0.1
logkernel(f,x), ((1-x)*log(1-x) + (1+x)*log(x+1) - 2)/π
```

```
(-0.6334313470059203, -0.6334313470059203)
```

Here is a plot of the solution:

```
Lf = z -> real((1-z)*log(z-1) + (1+z)*log(z+1) - 2)/π
Z = Lf.(xx' .+ im*yy)
contour(xx,yy,Z;ratio=1.0)
plot!(-1..1; color=:black, label="contour")
```


contour
2
1
0
−1
−2
−3
−2
−1
0
1
2
3
0.75
0.50
0.25
0
−0.25
−0.50

1.3.3 Example 3

Consider $f(x) = 1/\sqrt{1-x^2}$. We have

$$F(x) = \operatorname{acos} x = 2\operatorname{atan} \frac{\sqrt{1-x}}{\sqrt{1+x}}$$

where the second version can be verified by differentiation, using $\operatorname{atan}' x = 1/(1+x^2)$. One can determine M by indefinite integration of

$$\mathcal{C} f(z) = \frac{\mathrm{i}}{2\sqrt{z-1}\sqrt{z+1}}$$

but we prefer to start with an ansatz and verify the solution. Namely, consider

$$\phi(z) := \frac{\log(\sqrt{z-1} + \sqrt{z+1})}{\mathrm{i}}.$$

First, this is analytic off $(-\infty, 1]$, as for z in the upper-half plane we have $\sqrt{z-1}$ and $\sqrt{z+1}$ are in the upper-right quadrant: we never cross the branch cut of $\log z$. Similar argument holds for z in the lower-half plane. On $x \in (-\infty, -1]$ it has the jump

$$\begin{aligned}\phi_+(x) - \phi_-(x) &= \frac{\log(\mathrm{i}(\sqrt{1-x} + \sqrt{-x-1})) - \log(-\mathrm{i}(\sqrt{1-x} + \sqrt{-x-1}))}{\mathrm{i}} \\ &= 2\arg(\mathrm{i}(\sqrt{1-x} + \sqrt{-x-1})) = \pi\end{aligned}$$

For $x \in (-1,1)$ we have

$$\phi_+(x) - \phi_-(x) = \frac{\log(\mathrm{i}\sqrt{1-x} + \sqrt{1-x}) - \log(-\mathrm{i}\sqrt{1-x} + \sqrt{1-x})}{\mathrm{i}}$$
$$= 2\arg(\mathrm{i}\sqrt{1-x} + \sqrt{-x-1}) = 2\mathrm{atan}\,\frac{\sqrt{1-x}}{\sqrt{1+x}} = F(x).$$

Finally we have

$$\phi(z) = \frac{\log z}{2\mathrm{i}} - \mathrm{i}\log 2 + o(1), \qquad z \to \infty$$

We therefore have from Plemelj that

$$\mathcal{C}F(z) = \phi(z) - \frac{\log(z+1)}{2\mathrm{i}} + \mathrm{i}\log 2$$

and

$$Mf(z) = \frac{F(-1)}{\pi}\log(z+1) + 2\mathrm{i}\mathcal{C}F(z) = 2\log(\sqrt{z-1} + \sqrt{z+1}) - 2\log 2$$

Here is a plot of the solution:

```
Lf = z -> real(2log(sqrt(z-1) + sqrt(z+1)) - 2log(2))
Z = Lf.(xx' .+ im*yy)
contour(xx,yy,Z;ratio=1.0)
plot!(-1..1; color=:black, label="contour")
```



Note that it is clearly visible that it is constant on the contour, which follows from:

$$Lf(x) = \Re Mf(x) = 2\Re\log(\mathrm{i}\sqrt{1-x}+\sqrt{1+x}) - 2\log 2 = 2\log\sqrt{1-x+1+x} - 2\log 2$$
$$= -\log 2$$