Applied Complex Analysis (2021)

1 Solution Sheet 2

1.1 Problem 1

1.

We have

$$\sigma(A) \subseteq B(1,3) \cup B(2,3) \cup B(4,1)$$

where $B(z_0, r)$ is the ball of radius r around z_0.

Here's a depiction:

```
using ApproxFun, Plots, ComplexPhasePortrait, LinearAlgebra, DifferentialEquations
drawdisk!(z0, R) = plot!(θ-> real(z0) + R[1]*cos(θ), θ-> imag(z0) + R[1]*sin(θ), 0, 2π,
fill=(0,:red), α = 0.2, legend=false)

A = [1 2 -1; -2 2 1; 0 1 4]

λ = eigvals(A)
p = plot()
drawdisk!(1,3)
drawdisk!(2,3)
drawdisk!(4,1)
scatter!(complex.(λ); label="eigenvalues",ratio=1.0)
scatter!(complex.(diag(A)); label="diagonals")
p
```



2.

We get

$$\sigma(A) \subseteq B(1,2) \cup B(2,3) \cup B(4,2)$$

```
λ = eigvals(A)
p = plot()
drawdisk!(1,2)
drawdisk!(2,3)
drawdisk!(4,2)
scatter!(complex.(λ); label="eigenvalues")
scatter!(complex.(diag(A)); label="diagonals", ratio=1.0)
p
```



3.

Because the spectrum live in the intersection of the two estimates, the sharpest bound is

$$\sigma(A) \subseteq B(2,3)$$

```
λ = eigvals(A)
p = plot()
drawdisk!(2,3)
scatter!(complex.(λ); label="eigenvalues")
scatter!(complex.(diag(A)); label="diagonals",ratio=1.0)
p
```



Thus we can take $2+3\mathrm{e}^{\mathrm{i}\theta}$ as the contour.

1.2 Problem 2

1.

Note that in the scalar case $u'' = au$ we have the solution

$$u(t) = u_0 \cosh \sqrt{a}t + v_0 \frac{\sinh \sqrt{a}t}{\sqrt{a}}$$

Write

$$A = Q \begin{pmatrix} \lambda_1 & & \\ & \ddots & \\ & & \lambda_n \end{pmatrix} Q^\top$$

where $\lambda_k > 0$ and then the solution has the form, where γ is a contour surrounding the eigenvalues and to the right of zero:

$$
\begin{aligned}
\mathbf{u}(t) &= Q \begin{pmatrix} \cosh\sqrt{\lambda_1}t & & \\ & \ddots & \\ & & \cosh\sqrt{\lambda_n}t \end{pmatrix} Q^\top \mathbf{u}_0 + Q \begin{pmatrix} \frac{\sinh\sqrt{\lambda_1}t}{\sqrt{\lambda_1}} & & \\ & \ddots & \\ & & \frac{\sinh\sqrt{\lambda_n}t}{\sqrt{\lambda_n}} \end{pmatrix} Q^\top \mathbf{v}_0 \\
&= \frac{1}{2\pi\mathrm{i}} Q \begin{pmatrix} \oint_\gamma \frac{\cosh\sqrt{z}t\mathrm{d}z}{z-\lambda_1} & & \\ & \ddots & \\ & & \oint_\gamma \frac{\cosh\sqrt{z}t\mathrm{d}z}{z-\lambda_n} \end{pmatrix} Q^\top \mathbf{u}_0 \\
&\quad + \frac{1}{2\pi\mathrm{i}} Q \begin{pmatrix} \oint_\gamma \frac{\sinh\sqrt{z}t}{\sqrt{z}} \frac{\mathrm{d}z}{z-\lambda_1} & & \\ & \ddots & \\ & & \oint_\gamma \frac{\sinh\sqrt{z}t}{\sqrt{z}} \frac{\mathrm{d}z}{z-\lambda_n} \end{pmatrix} Q^\top \mathbf{v}_0 \\
&= \frac{1}{2\pi\mathrm{i}} \oint_\gamma \cosh\sqrt{z}t Q \begin{pmatrix} (z-\lambda_1)^{-1} & & \\ & \ddots & \\ & & (z-\lambda_n)^{-1} \end{pmatrix} Q^\top \mathbf{u}_0 \mathrm{d}z \\
&\quad + \frac{1}{2\pi\mathrm{i}} \oint_\gamma \frac{\sinh\sqrt{z}t}{\sqrt{z}} Q \begin{pmatrix} (z-\lambda_1)^{-1} & & \\ & \ddots & \\ & & (z-\lambda_n)^{-1} \end{pmatrix} Q^\top \mathbf{v}_0 \mathrm{d}z \\
&= \frac{1}{2\pi\mathrm{i}} \oint_\gamma \cosh\sqrt{z}t(zI-A)^{-1}\mathbf{u}_0\mathrm{d}z + \frac{1}{2\pi\mathrm{i}} \oint_\gamma \frac{\sinh\sqrt{z}t}{\sqrt{z}}(zI-A)^{-1}\mathbf{v}_0\mathrm{d}z
\end{aligned}
$$

Here we verify the formulae numerically:

```
n = 5
A = randn(n,n)
A = A+ A' + 10I

λ, Q = eigen(A)
λ
```

```
5-element Array{Float64,1}:
  5.433025054599263
  7.14656912761484
 10.157156268543757
 11.737040975203197
 14.225630632759346
```

```
norm(A - Q*Diagonal(λ)*Q')
```

```
1.9755263701875242e-14
```

Time-stepping solution:

```
u_0 = randn(n)
v_0 = randn(n)
uv = solve(ODEProblem((uv,_,t) -> [uv[n+1:end]; A*uv[1:n]], [u_0; v_0], (0.,2.));
reltol=1E-10);

t = 2.0
uv(t)[1:n]
```

```
5-element Array{Float64,1}:
 -837.5346933329671
```

```
 -514.9456697482184
 -211.89078614575558
 -393.65378907816563
  200.06771063707546
```

Solution via diagonalization:

```
Q*Diagonal(cosh.(sqrt.(λ) .* t))*Q'*u_0 + Q*Diagonal(sinh.(sqrt.(λ) .* t) ./
sqrt.(λ))*Q'*v_0

5-element Array{Float64,1}:
 -837.534693246912
 -514.9456696991397
 -211.89078611263412
 -393.6537890228899
  200.06771061234735
```

Solution via elliptic integrals. We chose the ellipse to surround all the spectrum of our particular A with eigenvalues:

```
periodic_rule(n) = 2π/n*(0:(n-1)), 2π/n*ones(n)

function ellipse_rule(n, a, b)
    θ = periodic_rule(n)[1]
    a*cos.(θ) + b*im*sin.(θ), 2π/n*(-a*sin.(θ) + im*b*cos.(θ))
end

function ellipse_f(f, A, n, z_0, a, b)
    z,w = ellipse_rule(n,a,b)
    z .+= z_0

    ret = zero(A)
    for j=1:n
        ret += w[j]*f(z[j])*inv(z[j]*I - A)
    end
    ret/(2π*im)
end

n = 50
ellipse_f(z -> cosh(sqrt(z)*t), A, n, 10.0, 7.0, 2.0)*u_0 +
    ellipse_f(z -> sinh(sqrt(z)*t)/sqrt(z), A, n, 10.0, 7.0, 2.0)*v_0

5-element Array{Complex{Float64},1}:
  -837.5352753313373 - 5.3232208872848765e-14im
   -514.945940243416 - 3.0596034914807286e-15im
 -211.89087791768256 + 2.7301183311249415e-14im
  -393.6542608371482 + 5.4515179803272316e-14im
  200.06781210361413 - 1.7445181448884913e-14im
```

2.

I put the restriction in because of the $\sqrt{z}$ term, which look like it is not analytic on $(-\infty, 0]$. However, this restriction was NOT necessary, since in fact $\cosh \sqrt{z}t$ and $\frac{\sinh \sqrt{z}t}{\sqrt{z}}$ are entire:

```
phaseplot(-6..6, -3..3, z -> cosh(sqrt(z)))
```


```
phaseplot(-3..3, -3..3, z -> sinh(sqrt(z))/sqrt(z))
```



This follows from Taylor series, though I prefer the following argument: we have

$$\cosh z = \frac{\mathrm{e}^z + \mathrm{e}^{-z}}{2}$$

hence $\cosh \mathrm{i}t = \cos t = \cosh -\mathrm{i}t$. Therefore, on the possible branch cut $\cosh \sqrt{z}$ is continuous (hence analytic):

$$\cosh \sqrt{x}_+ = \cosh \mathrm{i}\sqrt{|x|} = \cosh -\mathrm{i}\sqrt{|x|} = \cosh \sqrt{x}_-$$

Similarly,

$$\sinh z = \frac{\mathrm{e}^z - \mathrm{e}^{-z}}{2}$$

implies $\sinh \mathrm{i}t = \mathrm{i}\sin t = -\mathrm{i}\sin(-t) = -\sinh -\mathrm{i}t$, which gives us continuity:

$$\frac{\sinh \sqrt{x}_+}{\sqrt{x}_+} = \frac{\sinh \mathrm{i}\sqrt{|x|}}{\mathrm{i}\sqrt{|x|}} = \frac{\sinh(-\mathrm{i}\sqrt{|x|})}{-\mathrm{i}\sqrt{|x|}} = \frac{\sinh \sqrt{x}_-}{\sqrt{x}_-}$$

furthermore, they are both bounded at zero, hence analytic there too.

Here's a numerical example:

```
n = 5
A = randn(n,n)
λ, V = eigen(A)
λ

5-element Array{Complex{Float64},1}:
  -2.090123104000464 + 0.0im
 -0.9325315126496077 + 0.0im
  0.3856294421143206 - 1.746951187907767im
  0.3856294421143206 + 1.746951187907767im
  1.9562214750420424 + 0.0im

norm(A - V*Diagonal(λ)*inv(V))

5.674405190697963e-15

u_0 = randn(n)
v_0 = randn(n)
uv = solve(ODEProblem((uv,_,t) -> [uv[n+1:end]; A*uv[1:n]], [u_0; v_0], (0.,2.));
reltol=1E-10);

t = 2.0
uv(t)[1:n]

5-element Array{Float64,1}:
 -5.239799179580709
  2.2914869877453192
 -4.1323314014116805
 -0.4246539337453159
 -1.5572758569422693

V*Diagonal(cosh.(sqrt.(λ) .* t))*inv(V)*u_0 +
   V*Diagonal(sinh.(sqrt.(λ) .* t) ./ sqrt.(λ))*inv(V)*v_0
```

```
5-element Array{Complex{Float64},1}:
  -5.239799179393895 - 4.767176419984707e-16im
  2.2914869875654853 - 1.067962910696854e-15im
   -4.13233140119574 + 4.0670901795877945e-16im
 -0.4246539335944364 + 2.1951735326854606e-16im
 -1.5572758568465388 + 6.462385476705392e-16im
```

Here's the solution using an elliptic integral:

```
n = 100
ellipse_f(z -> cosh(sqrt(z)*t), A, n, 0.0, 3.0, 3.0)*u_0 +
    ellipse_f(z -> sinh(sqrt(z)*t)/sqrt(z), A, n, 0.0, 3.0, 3.0)*v_0
```

```
5-element Array{Complex{Float64},1}:
  -5.239799179393901 - 1.6300548875775391e-15im
   2.291486987565486 + 7.4764305196881e-18im
   -4.13233140119576 + 3.6994039358526127e-16im
 -0.4246539335944406 + 1.2033882116260187e-16im
 -1.5572758568465368 + 7.477362475410658e-16im
```

1.3 Problem 3

1.

We have

$$\frac{1}{n}\sum_{j=0}^{n-1} g(\theta_j) = \sum_{k=0}^{\infty} g_k \frac{1}{n}\sum_{j=0}^{n-1} \mathrm{e}^{\mathrm{i}k\theta_j}$$

Define the n-th root of unity as $\omega = e^{\frac{2\pi \mathrm{i}}{n}}$ (that is $\omega^n = 1$), and simplify

$$\sum_{j=0}^{n-1} \mathrm{e}^{\mathrm{i}k\theta_j} = \sum_{j=0}^{n-1} e^{\frac{2\pi jik}{n}} = \sum_{j=0}^{n-1} \omega^{kj} = \sum_{j=0}^{n-1} (\omega^k)^j$$

If k is a multiple of n, then $\omega^k = 1$, and this sum is equal to n. If k is not a multiple of n, use Geometric series:

$$\sum_{j=0}^{n-1} (\omega^k)^j = \frac{\omega^{nk} - 1}{\omega^k - 1} = \frac{1^k - 1}{\omega^k - 1} = 0$$

2.

From lecture 4, we have $|f_k| \leq M_r r^{-|k|}$ for any $1 \leq r < R$, where M_r is the supremum of f in an annulus $\{z : r^{-1} < |z| < r\}$. Thus from the previous part we have (using geometric series)

$$\left| \frac{1}{n}\sum_{j=0}^{n-1} g(\theta_j) - \frac{1}{2\pi}\int_0^{2\pi} g(\theta)\mathrm{d}\theta \right| \leq \sum_{K=1}^{\infty} |f_{Kn}| + |f_{-Kn}| \leq 2M \sum_{K=1}^{\infty} r^{-Kn} = 2M_r \frac{r^{-n}}{1 - r^{-n}}.$$

This is an upper bound that decays exponentially fast.

3.

Note that $f(z) = 2z/(4z - z^2 - 1)$ satisfies $f(e^{i\theta}) = g(\theta)$. This has two poles at $2 \pm \sqrt{3}$:

```
f = z -> 2z/(4z-z^2-1)
f(exp(0.1im)) - 1/(2-cos(0.1))
```

```
1.1102230246251565e-16 + 0.0im
```

Here's a phase plot showing the location of poles:

```
phaseplot(-5..5, -5..5, f)
```




```
2+sqrt(3), 2- sqrt(3), 1/(2+sqrt(3))
```

```
(3.732050807568877, 0.2679491924311228, 0.2679491924311227)
```

Note that $2 + \sqrt{3} = 1/(2 - \sqrt{3})$ so in the previous result, we take $R = 2 + \sqrt{3}$. For any $1 \le r < 2 + \sqrt{3}$ we have

$$M_r = \frac{2}{4 - r - r^{-1}}$$

Thus we get the upper bounds

$$\frac{4}{4 - r - r^{-1}} \frac{r^{-n}}{1 - r^{-n}}$$

Let's see how sharp it is:

```
periodic_rule(n) = 2π/n*(0:(n-1)), 2π/n*ones(n)
```

```julia
g = θ -> 1/(2 - cos(θ))

Q = sum(Fun(g, 0 .. 2π))

err = Float64[ (
            (θ, w) = periodic_rule(n);
            sum(w.*g.(θ)) - Q
        ) for n=1:30];

N = length(err)

scatter(abs.(err), yscale=:log10, label="true error", title="Trapezium error bounds",
legend=:bottomleft)
r = 1.1
scatter!(4/(4-r-inv(r)) .* r.^(-(1:N)) ./ (1 .- r.^(-(1:N))), label = "r = $r")
r = 3.0
scatter!(4/(4-r-inv(r)) .* r.^(-(1:N)) ./ (1 .- r.^(-(1:N))), label = "r = $r")
r = 2+sqrt(3)  - 0.01
scatter!(4/(4-r-inv(r)) .* r.^(-(1:N)) ./ (1 .- r.^(-(1:N))), label = "r = $r")
```